Supplementary File 1. SAS codes for the two macros.

The following pages provide the SAS code for the two macros created by our group, which are identified as “macro 1” and “macro 2”.
Both macros produce the same results but require different inputs and proceed differently. Factors influencing which macro potential users could use vary according to prior experience with SAS macro coding and ability to save novel data on the computer server on which running these macros. For users with access to storage space on the computer server and who appreciate receiving automated validations regarding potential coding errors, we recommend “macro 1”; for users who plan to run the macro on datasets with fewer calendar years and who have prior experience creating macro variables, we recommend “macro 2”.

/***********************************************************************
   MACRO 1															    

Consumer Price Index (CPI) 

Equation: Formula for adjusting for inflation using the CPI
			“CostYear2”= “CostYear1”* “CPIYear2”/ “CPIYear1”
Where :
“CPIYear1” represents the CPI for the year in which the cost was generated;
“CPIYear2” represents the CPI for the year you intend to use as the basis/reference for expressing your costs;
“CostYear1” represents the nominal cost at the time of incurring the expense;
“CostYear2” represents the cost adjusted for inflation;

PARAMETERS:
CPIreferenceFile= SAS datafile must include two variables: CPI value and year of reference for each CPI. One row by CPI.
yearCPI         = name of the variable that represents the year of reference for a CPI
CPI             = name of the variable that represents the CPI value

CostDataFile    = SAS datafile must include two variables: the cost and the date when the cost was generated
NominalCost     = name of the variable that represents the nominal cost (cost prior to adjustment for inflation)
DateNominalCost = name of the variable that represents the date when the nominal cost was generated

YearRef         = The year intended to be used as the basis for expressing costs
fileout         = The name of the datafile returned by the macro CPI
costadjusted    = The name of the variable that represents the costs once they are adjusted for inflation via the CPI. 
                  This variable will be a part of the datafile returned by the macro CPI.

*****************************************************************************************************************************************/




/**************************************************** MACRO 1 TO RUN ******************************************************************/

%MACRO CPI (CPIreferenceFile=, yearCPI=, CPI=, CostDataFile=, NominalCost=,  DateNominalCost=, YearRef=,  FileOut=, CostAdjusted=);

/* Verification that the 2 files pass in parameter exist */
%if %sysfunc(exist(&CPIreferenceFile)) %then %do;

	%if %sysfunc(exist(&CostDataFile)) %then %do;

			/*Finding the CPI associated to the year of reference chosen*/
			DATA CPI_REF;
			set &cpireferencefile.;
			IF &yearCPI.= &YearRef. THEN DO; CPIref=&CPI.; OUTPUT; END;
			Keep CPIref;
			run;

			data _null_; /*data step to check for number of records in dataset*/
			%global records;
			%let records=0;
			set CPI_REF end=last;
			if last then call symput("Records",put(_n_,8.)); /*_n_ is the observation number in sas dataset*/
			run;

			/*Verification of the existence of the value for the CPI for the year of reference*/
			 %if  &records ne 0 %then %DO ;

					/*Adding the CPI associated to the year of reference chosen to the file*/
					DATA CPI_CPI;
					merge CPI_REF &cpireferencefile.;
					IF _N_=1 THEN CPIreference=CPIref;
					retain CPIreference;
					yearmacroCPI=&yearCPI.;
					keep CPIreference yearmacroCPI &yearCPI. &CPI.;
					run;

					/*Create a variable who contain the year in which the cost was generated*/
					DATA CPI_CostData;
					set &CostDataFile.;
					yearmacroCPI= year(&DateNominalCost.);
					run;

					proc sort data=CPI_CPI;      by yearmacroCPI; run;
					proc sort data=CPI_CostData; by yearmacroCPI; run;

					/*Calculate the cost adjusted for inflation via the consumer price indices of the expenditure*/
					DATA &fileout.;
					merge CPI_CPI   CPI_CostData (in=A);
					by yearmacroCPI;
					IF A;
					IF &CPI ne 0 THEN &CostAdjusted. = round(&NominalCost.*CPIreference/ &CPI, 0.01);
					IF &CostAdjusted.= . THEN DO; 
							IF &CPI=.         THEN Error='No CPI found for the    ';
							IF &NominalCost=. THEN Error='No Nominal Cost';
							END;
					drop yearmacroCPI CPIreference &CPI;
					run;

					 /*deleting temporary files created by the macro CPI*/
					proc datasets lib=work nolist nodetails;delete CPI_REF CPI_CPI CPI_CostData;	quit;
				  %END;
			 %else %do;/*IF no CPI found for the Year ref a message will prompt*/
			      data _null_;
			      file print;
			      put #3 @10 "NO CPI reference find for the year &yearREf in the CPI refrence file";
			      run;
			   	  %end;
	  		%end;
	  %else %do;/*IF the file doesn't exist a message will prompt*/
	      	data _null_;
	        file print;
	        put #3 @10 "Data set &CostDataFile does not exist";
	      	run;
	   		%end;
	  %end;
  %else %do;/*IF the file doesn't exist a message will prompt*/
      	data _null_;
        file print;
        put #3 @10 "Data set &CPIreferenceFile does not exist";
      	run;
   		%end;
%MEND CPI ;



/*************************************************** MACRO 2 CODE TO BE USED BY ANALYST ******************************************************************/

%MACRO CPI (CPIreferenceFile=, yearCPI=, CPI=, CostDataFile=, NominalCost=,  DateNominalCost=, YearRef=,  FileOut=, CostAdjusted=);


/**************************************************** MACRO 2 TO RUN ******************************************************************/


/*
PARAMETERS:

CostDataFile    = SAS datafile must include two variables: the cost and the date when the cost was generated
NominalCost     = The name of the variable that represents the nominal cost (cost prior to the adjustment for inflation)
DateNominalCost = The name of the variable that represents the date when the nominal cost was generated

YearRef         = The year intended to be used as the basis for expressing costs
Fileout         = The name of the datafile returned by the macro CPI
Costadjusted    = The name of the variable that represents the costs once they are adjusted for inflation via the CPI. 
                  This variable will be a part of the datafile returned by the macro CPI.

Begin_y: First year of nominal cost data;
End_y: Last year of nominal cost data;
*/

%macro CPI2 (CostDataFile, Fileout, begin_y, end_y, YearRef, NominalCost, DateNominalCost);
	/* sets new dataset */
	data &Fileout;
	set &CostDataFile;
	/* error code if there is no CPI value for reference year (yearref)*/
	%if %symexist(cpi_&YearRef) %then %do; %put NOTE: CPI FOR REF YEAR EXISTS.; %end;
    %else %do; %put ERROR: NO CPI VALUE FOR THE REF YEAR (&YearRef); %abort cancel; %end;
	/* actualization of costs*/
	%do i = &begin_y %to &end_y;
    if year(&DateNominalCost)=&i then costadjusted = round((&NominalCost* &&cpi_&YearRef)/&&cpi_&i,0.01);
    %end;
	run;
%mend;


/*************************************************** MACRO 2 CODE TO BE USED BY ANALYST ******************************************************************/

* Create CPI macro variables for every year contained in your dataset and reference year. Change “year” text within the macro variable’s name for the numerical value of year within the dataset.
Example: 
%let cpi_2000 = 100;

%let cpi_year = ;
%let cpi_year = ;
%let cpi_year = ;
%let cpi_year = ;
%let cpi_year = ;
%let cpi_year = ;


%Macro CPI2 (CostDataFile, Fileout, begin_y, end_y, YearRef, NominalCost, DateNominalCost);
